U.S. Department of the Interior
Geological Survey

Desktop computer processing of coincident and central loop time domain electromagnetic data

by

Paul Raab and Frank Frischknecht

Open-File Report 83-240

1983

DISCLAIMER

Introduction

Field processing of geophysical data can lead to the early detection of interesting geologic features as well as the discovery of equipment or procedural problems. This report provides several BASIC programs, written for the Hewlett-Packard 85 desktop computer, which have proven useful for calculating the transient response of a homogeneous earth (forward problem) and for transforming time domain electromagnetic data (TDEM) to geologically more informative apparent resistivities (inverse problem). The programs presented herein were explicitly designed for use with an HP-85 desktop computer equipped with the Plotter/Printer, Input/Output, Advanced Programming, and Matrix ROMs, and 32K bytes of memory. External devices used include the HP-7225A (plotter), HP-82905A (printer), and the SIROTEM instrumentation system.

Typically, TDEM data has the form of a received voltage (V), normalized by a transmitted current (I), measured over a number of discrete time intervals following a current pulse (the center of this interval is the nominal time of measurement). Often, the design of equipment and preliminary planning of field surveys can be improved by predicting this transient response for a homogeneous earth model, presumably one which approximates an area of geologic interest. Two programs are provided to calculate such a transient response, for either the coincident ("CONFOR"; appendix I) or the central ("CENFOR"; appendix II) loop configurations, from a given set of earth and instrumentation system parameters. For example, input parameters of transmitter loop size, receiver moment (for the central loop configuration), sampling time, and half-space resistivity can be used to predict a range of received voltages.

In field surveys, the quantity V/I can vary by 5 or 6 orders of magnitude depending on the time range measured, the source configuration, and the geoelectrical section. Plotted curves of V/I vs. time tend to be rather featureless and difficult to interpret. Therefore, it is useful to transform a transient voltage to apparent resistivity by finding, for each measured time, the resistivity of a homogeneous half space which would produce the observed value of V/I. This transformation reduces the dynamic range of the data, normalizes it for loop size and time, and yields apparent resistivity vs. time curves. If lateral changes in the electrical properties of the local geologic section are small with respect to variations in the vertical section, and if the layers are thick, then the plot of apparent resistivity vs. time corresponds roughly to the plot of true resistivity vs. depth (Raiche and Spies, 1981).

Spies and Raiche (1980) published equations and a program for a hand held calculator to compute apparent resistivities for the coincident loop TDEM configuration. We have developed similar equations for the central loop or "in loop" configuration. However, for large loops, low resistivities, and early times the problem is slightly more difficult than for the coincident loop case. A data entry and storage program ("TDDATA"; appendix III) and an apparent resistivity calculation program ("TDRES"; appendix IV) are provided to process TDEM data using Spies and Raiche's algorithm for the coincident loop case and an algorithm detailed later in this paper for the central loop configuration.

There are limitations on the use of this apparent resistivity transformation. For example, with some electrical sections and times there is no homogeneous earth which will produce an equivalent value of V/I. Other transformations, which provide a good approximation to the apparent resistivity for either early or late times, are defined for all values of V/I and time but produce curves which cannot be viewed as soundings except over a limited temporal range.

Forward Problem

While the equations used in this report were derived for circular transmitting loops, they can, with little error, be used for square loops having an equivalent area (Spies and Raiche, 1980). To compute the transient response for a step function excitation of the coincident loop configuration, an expression given by Lee and Lewis (1974) can be written in the form

$$\frac{V}{I} = -\frac{4L\mu_o X^{1/2}}{t} \quad Y_1(X) \qquad (1)$$

where

$$Y_1(X) = \sum_{m=o}^{\infty} \frac{(-1)^m}{m!} \frac{(2m+2)! \quad X^{m+1}}{(m+1)!(m+2)!2(2m+5)}$$

$$X = \frac{\sigma\mu_o L^2}{4\pi t}$$

L = length of side of square loop

σ = conductivity of earth

t = time

$\mu_o = 4\pi \times 10^{-7}$

The series $Y_1(X)$ converges rapidly for small values of X. However, when X is large, many terms are needed for convergence. Large values of X would correspond to large loops, low resistivities, and early times. For X<=.5 the first ten terms of the series are used in the program "CONFOR" to solve for $Y_1(X)$. For .5<X<=1 the first 15 terms of the series are used. For 1<X<=3.2 the first 15 terms are summed and then Eulers transformation (Hildebrand, 1956) is applied to the next 10 terms and the result is added to the sum of the first terms. For 3.2<X<=5 the first 30 terms are summed and Eulers transformation is applied to the next 10 terms and added. Over most of the range 0<X<=5 the results are accurate to .2% or better. For X=4.5 the error is about .2%, but for X=5 it is about 2%. When X>5 a result is not calculated and the program gives a warning message.

For the quasi-static case, the vertical magnetic field about an electric dipole excited by a sinusoidal current is given by Wait (1961) as

$$H_z = \frac{I(\omega)dl}{2\pi\gamma^2 r^2}\left[3-(3+3\gamma r+\gamma^2 r^2)e^{-\gamma r}\right]\sin\phi \qquad (2)$$

where

$I(\omega)$ = current in dipole

dl = length of dipole

γ = $(i\sigma\mu_o\omega)^{1/2}$

ω = angular frequency

r = spacing between dipole and point of observation

ϕ = angle between direction of dipole and line between dipole and point of observation.

The field at the center of a circular loop of radius, a, may be determined by integrating this expression over an angle of 2π with $r = a$. The voltage induced in a small horizontal loop is

$$V(\omega) = i\mu_o\omega MH_z$$

where $M = nA$ = moment of the loop. To obtain the time domain voltage for a step function current $I\,u(t)$, replace $I(\omega)$ in equation (2) by its Laplace transform, I/s, replace $i\omega$ by s and take the inverse Laplace transform of the resulting expression (Wait, 1951). Ignoring the impulse response at $t = 0$, the voltage induced in a small receiving loop at the center of a square transmitting loop with sides L ($L = \sqrt{\pi}a$) is then

$$V(t) = \frac{MI\pi^{3/2}}{\sigma L^3}\left[3\,\mathrm{erf}(Z)-(3Z+2Z^3)\,\mathrm{erf}'(Z)\right]u(t) \qquad (3)$$

where

$$Z = \frac{L}{2}\left(\frac{\sigma\mu_o}{\pi t}\right)^{1/2} \quad \left(= X^{1/2}\right)$$

erf(Z) = error function with argument Z

erf'(Z) = derivative of error function $\left(= -\frac{2}{\sqrt{\pi}} e^{-Z^2}\right)$

For intermediate values of the argument, erf(Z) can readily be evaluated using approximations such as those given by Abramowitz and Stegun (1965). For small values of Z the approximation give by these authors is not sufficiently accurate since evaluation of equation (3) for small arguments involves the difference of two nearly equal terms. By expanding erf(Z) and erf'(Z) in series form and taking differences term by term a new series expression is obtained

$$\left[3\,\mathrm{erf}(Z)-(3Z+2Z^3)\,\mathrm{erf}'(Z)\right] = \frac{2Z^5}{\sqrt{\pi}}\left[\frac{4}{5} - \frac{4Z^2}{7} + \frac{2Z^4}{9} - \frac{2Z^6}{33} + \frac{Z^8}{78} - \frac{Z^{10}}{450} + \right.$$

$$\left. \frac{Z^{12}}{3{,}060} - \frac{Z^{14}}{40{,}320} + \frac{Z^{16}}{211{,}680} - \frac{Z^{18}}{2{,}086{,}560} + \frac{Z^{20}}{22{,}680{,}000} - \frac{Z^{22}}{269{,}438{,}400} + \ldots\right] \quad (4)$$

For Z<=.03, the program "CENFOR" uses the first two terms of equation (4) and for .03<Z<=.2 it uses the first six terms to evaluate erf(Z) and erf'(Z). The error function approximations give by Abramowitz and Stegun (eq. 7.1.26) are used in equation (3) to calculate V(t)/I when .2<Z<=4. When Z>4, erf(Z) is approximately one and the term involving erf'(Z) is very small. Hence, for Z>4 the following expression is used

$$\frac{V(t)}{I} = \frac{3M\pi^{3/2}}{\sigma L^3} = \frac{3\mu_o M \sqrt{\pi}}{4tLZ^2} \quad (5)$$

This program provides results accurate to about 0.1% over most of the range of Z values. Better results could be obtained for values of Z as large as about two if all 12 terms in equation (4) are used. However, this is slower than using equation (3) with the approximations for erf(Z).

Inverse Problem

The program "TDRES" uses the following equations to calculate apparent resistivities for the coincident and central loop cases. To obtain apparent resistivities from measurements of V(t)/I for the coincident loop configuration the following expressions were obtained by Spies and Raiche (1980) by reversion of the series in equation (1). Note that the correct value of the 7th term is the one listed in Spies and Raiches program and is not the one in their text (Spies, 1982).

$$X_1 = Z(1.710 + 2.381Z + 6.492Z^2 + 20.88Z^3 + 71.90Z^4 + 255.9Z^5 + 925.9Z^6 + 3,378Z^7 + 12,361Z^8 + 110,000Z^9)^2 \quad (6)$$

where

$$Z = Y_1^{2/3}$$

$$Y_1 = \frac{t}{2\mu_o L} \frac{V(t)}{I}$$

For $1.4<X_1<=5.69$, the following corrections, obtained by regression analysis, are applied

$$X = X_1 + .001635\ X_1^{4.892} \quad , \quad 1.4<X_1<=2.8$$

$$X = X_1 + .004018\ X_1^{4.01364} \quad , \quad 2.8<X_1<=5.69$$

If $X_1<1.4$, no correction is applied. If $X_1>5.69$ then X is not calculated and the program will print the message "NO SOLUTION". This allows X to be calculated for $X<=10$ with less than 1 percent error. After a value of X has been determined, an apparent resistivity is calculated using the following equation

$$\rho_a = \frac{\mu_o L^2}{4\pi t X}$$

The problem of determining apparent resistivities for the central loop configuration is a little more complex than for coincident loops. To deal with the inverse problem it is convenient to use a normalized quantity which does not explicitly contain σ.

$$Y(Z) = \frac{4\ t\ L}{\mu_o \sqrt{\pi}\ M} \frac{V(t)}{I} = \frac{1}{Z^2} \left[3\ \mathrm{erf}(Z) - (3Z+2Z^3)\ \mathrm{erf}'(Z)\right] u(t)$$

or in series form

$$Y(Z) = \frac{2Z^3}{\sqrt{\pi}} \left[\frac{4}{5} - \frac{4}{7} Z^2 + \frac{2}{9} Z^4 - \ldots\right] \quad (7)$$

From a plot of Y(Z) (Figure 1) it is seen that Z is a double-valued function of Y. In practice the left hand branch is of greatest interest but for large

loops, very low resistivities, and early times, Z may be on the right hand branch. It follows that for a given loop and time there are two values of resistivity which will give identical voltages.

The series in equation (7) can be formally reverted to obtain the expression

$$Z = 1.0772173U + .29761090U^3 - .1781113U^4 + .2622721U^5 - .3915436U^6 + .6381635U^7 - .8512796U^8 + 1.6251362U^9 - 2.325465U^{10} + \cdots \quad (8)$$

where

$$U = Y^{1/3}$$

Other expressions can be obtained depending on how equation (7) is manipulated. By an empirical process equation (8) was modified to give more accurate results over a wider range of parameters (equation (9)).

$$Z = 1.077217345U + .321U^3 - .14U^4 + .42U^5 - .22U^6 + .6U^7 - .86U^8 + .98U^9 - .8U^{10} + .68U^{11} - .32U^{12} + 1.52U^{13} - \cdots \quad (9)$$

When the smaller of the two values for Z is the proper one, the error in using equation (9) is less than 0.3% for Z<.5, less than 1% for Z<1 and less than 2% for Z<1.35. When Z is large it can be obtained by rearranging equation (5) to

$$Z \simeq \left(\frac{3}{Y}\right)^{1/2} \quad (10)$$

The program "TDRES" uses equation (9) when Z<=.5 and equation (10) when Z>3. When .5<Z<=3 , a value for Z is obtained by iterative solution of equation (7) using the Abramowitz and Stegun approximations for erf(Z). Errors in the iterative solution are usually less than 0.2% but when Y is near its maximum they may be as large as 1%. Having found Z, ρ_a is given by the expression

$$\rho_a = \frac{\mu_o L^2}{4\pi t Z^2} \quad (11)$$

Central loop data processing begins with the late time data and assumes that the last measured V(t)/I falls on the left hand branch (Figure 1). The value of Y is monitored and when it starts to decrease the program is switched to find solutions on the right hand branch only. However, for noisy data the program often switches prematurely to the wrong branch. Instead of trying to overcome this difficulty with a more complex algorithm it was decided to let the user choose the proper branch. By setting a switch in the program, computations can be restricted to the late time branch. In practice the early time branch is seldom encountered; for instance, if a 100 meter loop is used and the homogeneous half-space resistivity is 1 ohm-meter, the early time branch starts at about 0.384ms. The maximum value of Y for a homogeneous earth is about 0.701585; if the computed value is larger, there is no homogeneous earth which can produce the observed value of V(t)/I. In the program, if Y > 0.701585 a resistivity is not calculated and a warning message is printed.

Coincident Loop Configuration Forward Program

The program "CONFOR" will calculate the transient response (V/I) of a homogeneous half-space for the coincident loop configuration. After loading and running this program, the user must input the length of the transmitter loop side (meters), half-space resistivity (ohm-meters), and sample times (milliseconds) after transmitter turnoff. Up to 100 time values may be entered. When all desired time values have been entered, 'END LINE' is pressed without typing in a value. The quantity V/I is then calculated for the given model at the chosen sample times. Results are printed in microvolts/amp (Figure 2). If no solution is possible for a set of parameters, the message "NO SOLUTION" is printed in place of V/I.

Central Loop Configuration Forward Program

The program "CENFOR" will calculate the transient response (V/I) of a homogeneous half-space for the central loop configuration. After loading and running this program, the user must input the length of the transmitter loop side (meters), receiver moment (number of turns times loop area), half-space resistivity (ohm-meters), and sample times (milliseconds) after transmitter turnoff. Time values are entered and results are printed in the same way as described for the program "CONFOR".

TDEM Data Entry Program

"TDDATA" is an interactive data entry, editing, and storage program. Header records containing a project title, date, and station ID as well as parameters such as loop configuration (single loop, coincident loop, central loop or "in loop", and offset loop), transmitter loop size, and receiver moment (for central induction loop only) are entered for later use in the apparent resistivity calculation and processing program "TDRES". Normalized voltages are entered for up to 10 data sets with a maximum of 32 time intervals per data set. Header records and data arrays can be edited and data sets can be added or deleted. After the user is satisfied with the form of the data base, it can be stored on tape. Header records and data arrays can be printed on the HP-85.

After loading and starting the program, all eight special function keys of the HP-85 are labeled and can be used to select the following program options:

HELP - will print or display a brief help library explaining the basic operation of the program.

DATA - starts the header record and data entry subroutine. Up to 10 different data sets of a particular loop configuration may be combined into a single data base. Usually this data base represents several repeats from a single station. Header records containing identification and loop information are entered. The number of observations and V/I data for individual channels (from early to late times) are entered for each data set. Either a keyboard or tape entry mode may be selected. The tape entry mode is used to read SIROTEM data files created by the program "RDTAPE" (Bradley and Raab, 1983). The keyboard entry mode is used to manually enter transient data. When the user wishes to stop entering data for a particular set in the keyboard mode, the end line key is pressed without entering a value. After a preselected number of sets have been processed, program control returns to the main program driver to select another option.

READ - recovers previously combined and stored TDEM data from tape, usually to edit the header records and/or data values. A data file name is requested from the user and that data file is loaded and stored in memory. Note: this process will erase all previously entered data.

EDIT - executes the editing subroutine for both header records and individual data channels. If the header records are edited, each record is displayed and the user is requested to enter a new value for that record. If the end line key is pressed without entering a new value, the header record is left unchanged. Similarly, data values from each data set are displayed with a channel number corresponding to individual time intervals. The user may elect to edit, delete, or add selected data channels. Program control may be returned to the main driver without further editing by simply pressing the end line key until each data set has been examined.

ADD - adds more data sets of the same loop configuration to the existing data base. Header information is not requested but the number of observations and normalized transient voltages are entered from the keyboard for each additional data set.

DEL - deletes selected data sets, by number, from the data base. When the user wishes to return to the main driver, the end line key is pressed without selecting a data set. If an out of range value is selected, a new number is requested.

STORE - stores the header records and combined data sets. An output file name is requested. One must also specify whether the output file is new or old (i.e. whether there is another file with the same name on tape). If it is an old file name then the duplicate file on tape will be purged and replaced.

PRINT - Prints header records and data on the HP-85 thermal printer.

When each of the above subroutines have completed, control is always passed back to the main driver for selection of another program option. Data previously stored in memory will be replaced when the subroutines "DATA" or "READ" are initiated.

Apparent Resistivity Calculation Program

The program "TDRES" calculates apparent resistivities from normalized transient (microvolts/amp) data at each of 32 possible SIROTEM time channels. The transient data is input from tape (as stored by the data entry program "TDDATA" outlined above). Time values in milliseconds for each of the SIROTEM's 32 time channels are read from a "DATA" statement in the program. The apparent resistivity calculation algorithm for either the coincident or the central loop case (see earlier discussion) is used depending on header record information. Parameters required for the calculations, such as transmitter loop size and receiver moment, are also read from header records.

After loading and starting the program, one can select from several printing and/or plotting options (sample output illustrated in Figures 3a, 3b, 3c, and 3d). Defaults are given and may be selected by simply pressing end line. The user may also enable or disable the early time calculation for the central loop receiver case. If the early time branch is disabled all calculations will be forced to the late time, or left hand branch of Y(Z) (Figure 1). If the early time branch is enabled, the value of Y will be monitored, and if it starts to decrease, processing will switch to the early time calculation.

The program "TDRES" is designed to process a series of data files at one time. Data files are identified by a general file name and appended file numbers, both entered by the user. For example, consider three files named "TEST1", "TEST2", and "TEST3". The general file name "TEST" and the appended file numbers of 1, 2 or 3 would be used to select each file. If a general file name of "TEST1" were entered and no file numbers were selected, only the file "TEST1" would be processed. As a file is processed, results are printed and/or plotted on the selected devices. If the HP-7225A is used as a plotting device, paper must be loaded for each new plot. When this is required, the computer will beep and indicate that one should "LOAD PAPER AND CONTINUE". For all other options, processing should be automatic.

Suggestions for Converting to Other Systems

In general, all programs take advantage of several device specific BASIC language modifications. For example, the plotting subroutines make extensive use of Hewlett-Packard ROM defined keywords to drive the HP-85 CRT graphics display and 7225A plotter. Input/Output conventions are also device specific, as is to be expected. If other instrumentation than the SIROTEM is used to collect TDEM data, two specific changes should be made. In the data entry program "TDDATA" and the data processing program "TDRES", the 10 (data sets) by 32 (SIROTEM channel) element arrays for normalized voltages and calculated apparent resistivities and the 32 element array for time interval values (in milliseconds) should be redimensioned to represent the number of time intervals sampled. In "TDRES" the time interval values stored in the "DATA" statements (lines 240, 250 and 260 in this version) will also have to be changed for the particular data collection system used.

References

Abramowitz, M., and Stegun, I. H., (eds), Handback of Mathematical Functions (Appl. Math Ser. 55) NBS, Washington, D.C., Chap. 7.

Bradley, Jerry and Raab, Paul, 1983, SIROTEM II IEEE 488-1978 interface and controlling software: U.S. Geological Survey Open-File Report 83-247.

Hildebrand, F. B., 1956, Introduction to numerical analysis: New York, McGraw Hill 511 p.

Lee, T. and Lewis, R., 1974, Transient EM response of a large loop on a layered ground: Geophysical Prospecting, v. 22, no. 3 pp 430-444.

Raiche, A. P. and Spies, B. R., 1981, Coincident loop transient electromagnetic master curves for interpretation of two-layer earths: Geophysics, v. 46 no. 1, p. 53-64.

Spies, B. R. (1982), oral communication.

Spies, B. R. and Raiche, A.P., 1980, Calculation of apparent conductivity for the transient electromagnetic (coincident loop) method using an HP-67 calculator: Geophysics, v. 45 no. 7, p. 1197-1200.

Wait, J. R., 1961, The electromagnetic fields of a horizontal dipole in the presence of a conducting half-space: Can. Jour. Phy. v. 39, p. 1017-1028.

Wait, J. R., 1951, Oscillating magnetic dipole over a horizontally stratified earth: Can. Jour. Phy. v. 29, p. 577-592.



Figure 1. Plot of function Y(Z).

```
COINCIDENT LOOP FORWARD PROGRAM
XMTR LOOP SIDE(m) =  500
RESISTIVITY(ohm-m) =  10

 CH    TIME(msec)      microV/I
-------------------------------
  1    1.000E-001    NO SOLUTION
  2    5.000E-001    2.931E+005
  3    1.000E+000    1.252E+005
  4    5.000E+000    9.476E+003
  5    1.000E+001    2.247E+003
  6    5.000E+001    5.237E+001
  7    1.000E+002    9.588E+000
  8    5.000E+002    1.765E-001
  9    1.000E+003    3.130E-002
```

Figure 2. Printed output from the forward program "CONFOR".

```
INPUT FILE = MLEX03
TDRES SAMPLE OUTPUT
4/13/83
STATION 1
SETS 2
(3)CENTRAL INDUCTION LOOP
XMTR SIDE = 457
LOOP MOMENT = 11613

SET  1  CHAN =  32                    SET  2  CHAN =  32
STACKS  2048                          STACKS  2048

CHAN  TIME   microV/I     APP RES     CHAN  TIME   microV/I     APP RES
-----------------------------------   -----------------------------------
  1     .4  1.469E+004  5.77E+001       1     .4  1.463E+004  5.79E+00
  2     .8  4.077E+003  5.11E+001       2     .8  4.073E+003  5.11E+00
  3    1.2  1.939E+003  4.52E+001       3    1.2  1.928E+003  4.54E+00
  4    1.6  1.133E+003  4.13E+001       4    1.6  1.125E+003  4.16E+00
  5    2.0  7.813E+002  3.71E+001       5    2.0  7.793E+002  3.72E+00
  6    2.6  5.220E+002  3.18E+001       6    2.6  5.213E+002  3.18E+00
  7    3.4  3.407E+002  2.74E+001       7    3.4  3.397E+002  2.75E+00
  8    4.2  2.429E+002  2.44E+001       8    4.2  2.418E+002  2.45E+00
  9    5.0  1.835E+002  2.21E+001       9    5.0  1.828E+002  2.22E+00
 10    5.8  1.449E+002  2.03E+001      10    5.8  1.446E+002  2.04E+00
 11    7.0  1.058E+002  1.85E+001      11    7.0  1.053E+002  1.85E+00
 12    8.6  7.220E+001  1.71E+001      12    8.6  7.217E+001  1.71E+00
 13   10.2  5.187E+001  1.61E+001      13   10.2  5.193E+001  1.61E+00
 14   11.8  3.827E+001  1.56E+001      14   11.8  3.797E+001  1.57E+00
 15   13.4  2.900E+001  1.52E+001      15   13.4  2.905E+001  1.52E+00
 16   15.8  1.986E+001  1.50E+001      16   15.8  1.971E+001  1.51E+00
 17   19.0  1.248E+001  1.51E+001      17   19.0  1.248E+001  1.51E+00
 18   22.2  8.277E+000  1.54E+001      18   22.2  8.297E+000  1.54E+00
 19   25.4  5.717E+000  1.59E+001      19   25.4  5.637E+000  1.60E+00
 20   28.6  4.043E+000  1.65E+001      20   28.6  3.997E+000  1.66E+00
 21   33.4  2.701E+000  1.67E+001      21   33.4  2.755E+000  1.65E+00
 22   39.8  1.675E+000  1.72E+001      22   39.8  1.632E+000  1.75E+00
 23   46.2  1.167E+000  1.71E+001      23   46.2  1.211E+000  1.67E+00
 24   52.6  8.700E-001  1.68E+001      24   52.6  8.027E-001  1.77E+00
 25   59.0  6.583E-001  1.67E+001      25   59.0  7.093E-001  1.59E+00
 26   68.6  4.703E-001  1.63E+001      26   68.6  4.760E-001  1.61E+00
 27   81.4  3.350E-001  1.53E+001      27   81.4  3.223E-001  1.58E+00
 28   94.2  2.240E-001  1.58E+001      28   94.2  2.253E-001  1.57E+00
 29  107.0  1.663E-001  1.56E+001      29  107.0  1.690E-001  1.54E+00
 30  119.8  1.347E-001  1.48E+001      30  119.8  1.347E-001  1.48E+00
 31  139.0  7.867E-002  1.66E+001      31  139.0  8.433E-002  1.58E+00
 32  164.6  3.567E-002  2.12E+001      32  164.6  3.200E-002  2.28E+00
```

Figure 3a. Printed results of the program "TDRES" from the HP-85 printer.





Figure 3b. Plots of V/I vs. time and apparent resistivity vs. time drawn on the HP-85.

INPUT FILE = MLEX03
TDRES SAMPLE OUTPUT 4/13/83 STATION 1
(3)CENTRAL INDUCTION LOOP XMTR SIDE = 457 LOOP MOMENT = 11613

		SET 1 CHAN = 32 STACKS = 2048		SET 2 CHAN = 32 STACKS = 2048	
CH	T(ms)	microV/I	APP. RES.	microV/I	APP. RES.
1	.4	1.468E+004	5.76E+001	1.463E+004	5.79E+001
2	.8	4.076E+003	5.10E+001	4.073E+003	5.10E+001
3	1.2	1.939E+003	4.51E+001	1.927E+003	4.54E+001
4	1.6	1.133E+003	4.13E+001	1.125E+003	4.15E+001
5	2	7.813E+002	3.71E+001	7.793E+002	3.71E+001
6	2.6	5.22 E+002	3.17E+001	5.213E+002	3.18E+001
7	3.4	3.406E+002	2.74E+001	3.396E+002	2.74E+001
8	4.2	2.429E+002	2.43E+001	2.418E+002	2.44E+001
9	5	1.835E+002	2.21E+001	1.828E+002	2.21E+001
10	5.8	1.449E+002	2.03E+001	1.446E+002	2.03E+001
11	7	1.057E+002	1.84E+001	1.052E+002	1.85E+001
12	8.6	7.22 E+001	1.70E+001	7.216E+001	1.70E+001
13	10.2	5.186E+001	1.61E+001	5.193E+001	1.60E+001
14	11.8	3.826E+001	1.55E+001	3.796E+001	1.56E+001
15	13.4	2.9 E+001	1.52E+001	2.904E+001	1.52E+001
16	15.8	1.986E+001	1.50E+001	1.971E+001	1.50E+001
17	19	1.247E+001	1.51E+001	1.248E+001	1.51E+001
18	22.2	8.276E+000	1.54E+001	8.296E+000	1.54E+001
19	25.4	5.716E+000	1.58E+001	5.636E+000	1.60E+001
20	28.6	4.043E+000	1.64E+001	3.996E+000	1.65E+001
21	33.4	2.701E+000	1.66E+001	2.755E+000	1.64E+001
22	39.8	1.674E+000	1.71E+001	1.632E+000	1.74E+001
23	46.2	1.167E+000	1.70E+001	1.210E+000	1.66E+001
24	52.6	8.7 E-001	1.67E+001	8.026E-001	1.76E+001
25	59	6.583E-001	1.66E+001	7.093E-001	1.58E+001
26	68.6	4.703E-001	1.62E+001	4.76 E-001	1.61E+001
27	81.4	3.35 E-001	1.53E+001	3.223E-001	1.57E+001
28	94.2	2.24 E-001	1.57E+001	2.253E-001	1.56E+001
29	107	1.663E-001	1.55E+001	1.69 E-001	1.53E+001
30	119.8	1.346E-001	1.48E+001	1.346E-001	1.48E+001
31	139	7.866E-002	1.65E+001	8.433E-002	1.58E+001
32	164.6	3.566E-002	2.12E+001	3.2 E-002	2.28E+001

Figure 3c. HP-82905A printer output from the program "TDRES".



Figure 3d. Plot of V/I (solid line) and apparent resistivity (dashed line) vs. time. Drawn on the HP-7225A.

Appendix I. Program listing for "CONFOR".

```
10 ! ******** CONFOR **********
20 ! * FORWARD CALCULATION    *
30 ! * OF THE TDEM RESPONSE   *
40 ! * OF A HOMOGENEOUS EARTH *
50 ! * USING THE COINCIDENT   *
60 ! * LOOP CONFIGURATION.    *
70 ! * INPUT OF PARAMETERS    *
80 ! * FROM THE HP-85 KEYS.   *
90 ! **************************
100 ! BY FRANK FRISCHKNECHT AND PAUL RAAB - USGS
110 ! MODIFIED 4/13/83
120 OPTION BASE 1
130 CRT IS 1 @ PRINTER IS 2
140 CLEAR
150 DIM C(40),T(100),V(100),E(10),U(10),A$[11],Z$[32]
160 A$="NO SOLUTION"
170 !
180 ! SERIES COEFFICIENTS
190 !
200 DATA .1,.14285714286,.13888888889
210 DATA .10606060606,.067307692308,.036666666667
220 DATA .017524509804,7.4665831245E-3,2.8710789872E-3
230 DATA .001006201595,3.2399691358E-4,9.6502577213E-5
240 DATA 2.6740246149E-5,6.9272498708E-6,1.6849560698E-6
250 DATA 3.8626555952E-7,8.3734369604E-8,1.7216163949E-8
260 DATA 3.366249943E-9,6.2745627543E-10,1.1173792057E-10
270 DATA 1.9048753473E-11,3.1144181814E-12,4.8917340642E-13
280 DATA 7.39262423E-14,1.0764855646E-14,1.5124134475E-15
290 DATA 2.0526943992E-16,2.6944618518E-17,3.4243933135E-18
300 DATA 4.2179306103E-19,5.0400273849E-20,5.8475381663E-21
310 DATA 6.5930538243E-22,7.2296917102E-23,7.7161373766E-24
320 DATA 8.0211612621E-25,8.126916724E-26,8.0305449043E-27
330 DATA 7.7439147726E-28
340 MAT READ C
350 !
360 ! GET MODEL PARAMETERS
370 !
380 DISP "ENTER LENGTH OF XMTR SIDE (m)"
390 INPUT L
400 DISP @ DISP "ENTER HALF-SPACE RESISTIVITY
(ohm-meters)"
410 INPUT R
420 N=0
430 FOR I=1 TO 100
440 DISP @ DISP "ENTER CHANNEL ";I;" TIME (msec)"
450 DISP "HIT ONLY END LINE TO QUIT" @ INPUT Z$
460 IF Z$="" THEN 510
470 T(I)=VAL(Z$)
```

```
480 N=N+1
490 NEXT I
500 !
510 ! PROCESSING LOOP
520 !
530 CLEAR
540 FOR I=1 TO N
550 X=.0001*L*L/(R*T(I))
560 IF X>5 THEN V(I)=0 @ GOTO 1370
570 ! CALCULATE FIRST TEN TERMS OF THE SERIES Y(X)
580 !
590 Y=X*(C(1)-X*(C(2)-X*(C(3)-X*(C(4)-X*C(5)))))
600 Y=Y-X^6*(C(6)-X*(C(7)-X*(C(8)-X*(C(9)-X*C(10)))))
610 IF X<=.5 THEN 1340
620 !
630 ! CALCULATE FIRST 15 TERMS OF THE SERIES Y(X)
640 !
650 Y=Y+X^11*(C(11)-X*(C(12)-X*(C(13)-X*(C(14)-X*C(15)))))
660 IF X<=1 THEN 1340
670 IF X>1 AND X<=3.2 THEN 700
680 IF X>3.2 AND X<=5 THEN 1010
690 !
700 ! USE THE FIRST 15 TERMS OF THE SERIES Y(X)
710 ! AND EULER'S TRANSFORMATION ON THE NEXT TEN TERMS
720 !
730 E(1)=C(16)
740 E(2)=C(17)*X
750 E(3)=C(18)*X*X
760 E(4)=C(19)*X^3
770 E(5)=C(20)*X^4
780 E(6)=C(21)*X^5
790 E(7)=C(22)*X^6
800 E(8)=C(23)*X^7
810 E(9)=C(24)*X^8
820 E(10)=C(25)*X^9
830 U(1)=E(1)
840 U(2)=E(2)-E(1)
850 U(3)=E(3)-2*E(2)+E(1)
860 U(4)=E(4)-3*E(3)+3*E(2)-E(1)
870 U(5)=E(5)-4*E(4)+6*E(3)-4*E(2)+E(1)
880 U(6)=E(6)-5*E(5)+10*E(4)-10*E(3)+5*E(2)-E(1)
890 U(7)=E(7)-6*E(6)+15*E(5)-20*E(4)+15*E(3)-6*E(2)+E(1)
900 U(8)=E(8)-7*E(7)+21*E(6)-35*E(5)
910 U(8)=U(8)+35*E(4)-21*E(3)+7*E(2)-E(1)
920 U(9)=E(9)-8*E(8)+28*E(7)-56*E(6)+70*E(5)
930 U(9)=U(9)-56*E(4)+28*E(3)-8*E(2)+E(1)
940 U(10)=E(10)-9*E(9)+36*E(8)-84*E(7)+126*E(6)
950 U(10)=U(10)-126*E(5)+84*E(4)-36*E(3)+9*E(2)-E(1)
960 Y=Y-X^16*(U(1)/2-U(2)/4+U(3)/8-U(4)/16)
970 Y=Y-X^16*(U(5)/32-U(6)/64+U(7)/128)
980 Y=Y-X^16*(-(U(8)/256)+U(9)/512-U(10)/1024)
```

```
990 GOTO 1340
1000 !
1010 ! USE THE FIRST 30 TERMS OF THE SERIES Y(X)
1020 ! AND EULER'S TRANSFORMATION ON THE NEXT TEN TERMS
1030 !
1040 Y=Y-X^16*(C(16)-X*(C(17)-X*(C(18)-X*(C(19)-X*C(20)))))
1050 Y=Y+X^21*(C(21)-X*(C(22)-X*(C(23)-X*(C(24)-X*C(25)))))
1060 Y=Y-X^26*(C(26)-X*(C(27)-X*(C(28)-X*(C(29)-X*C(30)))))
1070 E(1)=C(31)*X^15
1080 E(2)=C(32)*X^16
1090 E(3)=C(33)*X^17
1100 E(4)=C(34)*X^18
1110 E(5)=C(35)*X^19
1120 E(6)=C(36)*X^20
1130 E(7)=C(37)*X^21
1140 E(8)=C(38)*X^22
1150 E(9)=C(39)*X^23
1160 E(10)=C(40)*X^24
1170 U(1)=E(1)
1180 U(2)=E(2)-E(1)
1190 U(3)=E(3)-2*E(2)+E(1)
1200 U(4)=E(4)-3*E(3)+3*E(2)-E(1)
1210 U(5)=E(5)-4*E(4)+6*E(3)-4*E(2)+E(1)
1220 U(6)=E(6)-5*E(5)+10*E(4)-10*E(3)+5*E(2)-E(1)
1230 U(7)=E(7)-6*E(6)+15*E(5)-20*E(4)+15*E(3)-6*E(2)+E(1)
1240 U(8)=E(8)-7*E(7)+21*E(6)-35*E(5)
1250 U(8)=U(8)+35*E(4)-21*E(3)+7*E(2)-E(1)
1260 U(9)=E(9)-8*E(8)+28*E(7)-56*E(6)+70*E(5)
1270 U(9)=U(9)-56*E(4)+28*E(3)-8*E(2)+E(1)
1280 U(10)=E(10)-9*E(9)+36*E(8)-84*E(7)+126*E(6)
1290 U(10)=U(10)-126*E(5)+84*E(4)-36*E(3)+9*E(2)-E(1)
1300 Y=Y+X^16*(U(1)/2-U(2)/4+U(3)/8-U(4)/16)
1310 Y=Y+X^16*(U(5)/32-U(6)/64+U(7)/128)
1320 Y=Y+X^16*(-(U(8)/256)+U(9)/512-U(10)/1024)
1330 !
1340 ! CALCULATE TRANSIENT
1350 !
1360 V(I)=16*PI*L*L/(T(I)*SQR(R*T(I)))*Y
1370 NEXT I
1380 !
1390 ! PRINT RESULTS
1400 !
1410 CRT OFF
1420 PRINT @ PRINT
1430 PRINT "COINCIDENT LOOP FORWARD PROGRAM"
1440 PRINT "XMTR LOOP SIDE(m) = ";L
1450 PRINT "RESISTIVITY(ohm-m) = ";R
1460 PRINT
1470 PRINT " CH   TIME(msec)    microV/I"
1480 PRINT "-------------------------------"
1490 FOR I=1 TO N
```

```
1500 IF V(I)>0 THEN PRINT USING 1510 ; I,T(I),V(I)
1510 IMAGE 1X,2D,3X,1D.3DE,3X,1D.3DE
1520 IF V(I)=0 THEN PRINT USING 1530 ; I,T(I),A$
1530 IMAGE 1X,2D,3X,1D.3DE,3X,11A
1540 NEXT I
1550 PRINT @ PRINT @ PRINT @ PRINT @ PRINT
1560 CRT ON
1570 CLEAR @ DISP "NORMAL END"
1580 END
```

Appendix II. Program listing for "CENFOR".

```
10 ! ******** CENFOR **********
20 ! * FORWARD CALCULATION    *
30 ! * OF THE TDEM RESPONSE   *
40 ! * OF A HOMOGENEOUS EARTH *
50 ! * USING THE CENTRAL LOOP *
60 ! * CONFIGURATION. INPUT   *
70 ! * OF PARAMTERS FROM THE  *
80 ! * HP-85 KEYBOARD.        *
90 ! **************************
100 ! BY FRANK FRISCHKNECHT AND PAUL RAAB - USGS
110 ! MODIFIED 4/13/83
120 OPTION BASE 1
130 CRT IS 1 @ PRINTER IS 2
140 CLEAR
150 DIM T(100),V(100),Z$[32]
160 !
170 ! ERROR FUNCTION DEFINITION
180 ! FOR CENTRAL LOOP CASE .2<Z<=4
190 ! FROM ABRAMOWITZ & STEGUN, 1965
200 !
210 DEF FNF(Z)
220 U=1/(1+.3275911*Z)
230 U1=U*(.254829592+U*(-.284496736+U*1.421413741))
240 U1=U1+U^4*(-1.453152027+U*1.061405429)
250 U1=3-(3*U1+(3*Z+2*Z^3)*2/SQR(PI))*EXP(-(Z^2))
260 FNF=Z^(-2)*U1
270 FN END
280 !
290 ! GET MODEL PARAMETERS
300 !
310 DISP "ENTER TRANSMITTER LOOP SIDE     (meters)"
320 INPUT L
330 DISP @ DISP "ENTER RECEIVER MOMENT (n*A)"
340 INPUT M
350 DISP @ DISP "ENTER HALF-SPACE RESISTIVITY"
360 DISP "(ohm-meters)" @ INPUT R
370 N=0
380 FOR I=1 TO 100
390 DISP @ DISP "ENTER CHANNEL ";I;" TIME (msec)"
400 DISP "HIT ONLY END LINE TO QUIT" @ INPUT Z$
410 IF Z$="" THEN 460
420 T(I)=VAL(Z$)
430 N=N+1
440 NEXT I
450 !
460 ! PROCESSING LOOP
470 !
480 CLEAR
```

```
490 FOR I=1 TO N
500 Z=.01*L/SQR(R*T(I))
510 IF Z<=.03 THEN 550
520 IF Z>.03 AND Z<=.2 THEN 590
530 IF Z>.2 AND Z<=4 THEN 640
540 IF Z>4 THEN 680
550 ! USE FIRST TWO TERMS
560 S=Z^3*(4/5-Z*Z*4/7)
570 V(I)=200*PI*M*S/(T(I)*L)
580 GOTO 700
590 ! USE THE FIRST SIX TERMS
600 S=Z^3*(4/5-Z*Z*(4/7+Z*Z*(2/9-Z*Z*2/33)))
610 S=S+Z^11*(1/78-Z*Z/450)
620 V(I)=200*PI*M*S/(T(I)*L)
630 GOTO 700
640 ! USE NBS ERROR FUNCTION APPROXIMATION
650 F=FNF(Z)
660 V(I)=100*M*PI^(3/2)*F/(T(I)*L)
670 GOTO 700
680 ! USE EARLY TIME APP
690 V(I)=300*M*PI^(3/2)/(T(I)*L*Z*Z)
700 NEXT I
710 !
720 ! PRINT RESULTS
730 !
740 CRT OFF
750 PRINT @ PRINT
760 PRINT "CENTRAL LOOP FORWARD PROGRAM"
770 PRINT "XMTR LOOP SIDE(m) = ";L
780 PRINT "RCVR LOOP MOMENT = ";M
790 PRINT "RESISTIVITY(ohm-m) = ";R
800 PRINT
810 PRINT " CH   TIME(msec)    microV/I"
820 PRINT "-------------------------------"
830 FOR I=1 TO N
840 PRINT USING 850 ; I,T(I),V(I)
850 IMAGE 1X,2D,3X,1D.3DE,3X,1D.3DE
860 NEXT I
870 PRINT @ PRINT @ PRINT @ PRINT
880 CRT ON
890 CLEAR @ DISP "NORMAL END"
900 END
```

Appendix III. Program listing for "TDDATA".

```
10 ! ******** TDDATA **********
20 ! * PROGRAM TO ENTER,EDIT, *
30 ! * AND STORE TDEM DATA    *
40 ! * AND LOOP PARAMETERS    *
50 ! * FOR INPUT TO 'TDRES'.  *
60 ! * PROGRAM OPTIONS ARE    *
70 ! * EXPLAINED ON-LINE BY   *
80 ! * A HELP FILE.           *
90 ! *************************
100 ! BY PAUL RAAB - USGS
110 ! MODIFIED 4/13/83
120 OPTION BASE 1
130 CRT IS 1 @ PRINTER IS 2
140 INTEGER O(10),N(10)
150 SHORT V(10,32),VO(32)
160 DIM A$[32],B$[8],C$[32],D$[8],E$[25],F$[18],G$[25]
170 DIM N$[17],NO$[17],N1$[96]
180 DIM H1$[32],H2$[32],H3$[32],H4$[32],Z$[32]
190 !
200 ! OPTION SELECT
210 !
220 ON KEY# 1,"HELP" GOSUB 3290
230 ON KEY# 2,"DATA" GOSUB 340
240 ON KEY# 3,"READ" GOSUB 1230
250 ON KEY# 4,"EDIT" GOSUB 1410
260 ON KEY# 5,"ADD" GOSUB 2270
270 ON KEY# 6,"DEL" GOSUB 2480
280 ON KEY# 7,"STORE" GOSUB 2700
290 ON KEY# 8,"PRINT" GOSUB 2970
300 CLEAR @ SFLAG 1 @ KEY LABEL
310 DISP "WAITING FOR KEY SELECTION"
320 IF FLAG(1) THEN 320 ELSE 300 ! WAIT FOR KEY HIT
330 !
340 ! DATA ENTRY
350 !
360 CLEAR @ CFLAG 1
370 DISP "ENTER PROJECT TITLE - 32 max" @ INPUT A$
380 DISP @ DISP "ENTER DATE - 8 max" @ INPUT B$
390 DISP @ DISP "ENTER STATION ID - 24 max"
400 C$[1,8]="STATION " @ INPUT C$[9]
410 DISP @ DISP "ENTER XMTR-RCVR LOOP
CONFIGURATION"
420 DISP "1=SINGLE LOOP"
430 DISP "2=COINCIDENT LOOP"
440 DISP "3=CENTRAL INDUCTION LOOP"
450 DISP "4=OFFSET LOOP"
460 INPUT E
470 IF E=1 THEN E$="(1)SINGLE LOOP"
```

```
480 IF E=2 THEN E$="(2)COINCIDENT LOOP"
490 IF E=3 THEN E$="(3)CENTRAL INDUCTION LOOP"
500 IF E=4 THEN E$="(4)OFFSET LOOP"
510 DISP @ DISP "ENTER LENGTH OF XMTR SIDE (m)"
520 F$[1,12]="XMTR SIDE = " @ INPUT F$[13]
530 IF E=3 THEN 570
540 DISP @ DISP "ENTER # OF XMTR LOOP TURNS"
550 G$[1,14]="XMTR TURNS  = " @ INPUT G$[15]
560 GOTO 600
570 DISP @ DISP "ENTER RCVR LOOP MOMENT"
580 DISP "(AREA x # OF TURNS)"
590 G$[1,14]="RCVR MOMENT = " @ INPUT G$[15]
600 MAT O=ZER@ MAT N=ZER@ MAT V=ZER
610 CLEAR
620 DISP "SELECT DATA ENTRY OPTION"
630 DISP "0=KEYBOARD ENTRY"
640 DISP "1=TAPE ENTRY VIA 'RDTAPE' FILES"
650 DISP "DEFAULT=0" @ INPUT Z$
660 IF Z$="" THEN 1020
670 IF Z$#"1" THEN 1020
680 !
690 ! MASS STORAGE ENTRY
700 !
710 CLEAR
720 DISP "ENTER GENERAL INPUT FILE NAME" @ INPUT NO$
730 DISP @ DISP "SELECT FILE NUMBERS TO PROCESS"
740 LINPUT "(e.g. 1,2,3,4 <CR/LF>)          ?",N1$
750 P1=1 @ I,N,P2=0
760 IF LEN(N1$)=0 THEN N=1 @ N$=TRIM$(NO$) @ GOTO 820
770 P2=POS(N1$[P1],",")+P1-2
780 I=I+1
790 N=N+1
800 IF P2>P1 THEN NO=VAL(N1$[P1,P2]) ELSE NO=VAL(N1$[P1])
810 N$=TRIM$(NO$)&VAL$(NO)
820 ON ERROR GOTO 3840
830 ASSIGN# 1 TO N$
840 CRT OFF
850 READ# 1 ; H1$,H2$,H3$,H4$
860 N(I)=VAL(H3$[17])
870 FOR J=1 TO N(I)
880 READ# 1 ; I$
890 V=VAL(I$[4,7])
900 E=VAL(I$[9,9])
910 V(I,J)=V*10^E*10^-3
920 IF V(I,J)=0 THEN N(I)=N(I)-1
930 NEXT J
940 READ# 1 ; I$
950 O(I)=VAL(I$[4,7])
960 ASSIGN# 1 TO *
970 IF P2>=P1 THEN P1=P2+2 @ GOTO 770
980 D$[1,5]="SETS " @ D$[6]=VAL$(N)
```

```
990 OFF ERROR @ CRT ON
1000 RETURN
1010 !
1020 ! KEYBOARD ENTRY
1030 !
1040 CLEAR
1050 DISP "ENTER NUMBER OF DATA SETS       TO PROCESS"
1060 D$[1,5]="SETS " @ INPUT D$[6]
1070 N=VAL(D$[6])
1080 FOR I=1 TO N
1090 CLEAR
1100 DISP "PROCESS DATA SET ";I
1110 DISP @ DISP "ENTER NUMBER OF STACKS" @ INPUT O(I)
1120 FOR J=1 TO 32
1130 DISP @ DISP "ENTER CHAN ";J;" microvolts/amp"
1140 DISP "ENTER ZERO TO SKIP CHANNEL"
1150 DISP "PRESS ONLY END LINE TO QUIT" @ INPUT Z$
1160 IF Z$="" THEN 1200
1170 V(I,J)=VAL(Z$)
1180 IF V(I,J)#0 THEN N(I)=N(I)+1
1190 NEXT J
1200 NEXT I
1210 RETURN
1220 !
1230 ! DATA RECOVERY
1240 !
1250 CLEAR @ CFLAG 1
1260 DISP "ENTER INPUT FILE NAME" @ INPUT N$
1270 ON ERROR GOTO 3840
1280 ASSIGN# 1 TO N$
1290 CRT OFF
1300 READ# 1 ; A$,B$,C$,D$,E$,F$,G$
1310 N=VAL(D$[6])
1320 FOR I=1 TO N
1330 READ# 1 ; N(I),O(I),V0()
1340 MAT V(I,1:32)=V0
1350 NEXT I
1360 ASSIGN# 1 TO *
1370 OFF ERROR
1380 CRT ON
1390 RETURN
1400 !
1410 ! EDIT HEADER & DATA
1420 !
1430 CLEAR @ CFLAG 1
1440 DISP A$
1450 DISP B$
1460 DISP C$
1470 DISP D$
1480 DISP E$
1490 DISP F$
```

```
1500 DISP G$
1510 DISP @ DISP "EDIT HEADER? Y/N - DEF IS NO" @ INPUT Z$
1520 IF Z$="" THEN 1880
1530 IF UPC$(Z$[1,1])#"Y" THEN 1880
1540 CLEAR
1550 DISP A$
1560 DISP "ENTER NEW PROJECT TITLE - 32 max"
1570 DISP "PRESS ONLY END LINE TO KEEP SAME" @ INPUT Z$
1580 IF Z$#"" THEN A$=Z$
1590 DISP @ DISP B$
1600 DISP "ENTER NEW DATE - 8 max"
1610 DISP "PRESS ONLY END LINE TO KEEP SAME" @ INPUT Z$
1620 IF Z$#"" THEN B$=Z$
1630 DISP @ DISP C$
1640 DISP "ENTER NEW STATION ID - 24 max"
1650 DISP "PRESS ONLY END LINE TO KEEP SAME" @ INPUT Z$
1660 IF Z$#"" THEN C$[9]=Z$
1670 DISP @ DISP E$
1680 DISP "ENTER NEW LOOP CONFIGURATION"
1690 DISP "PRESS ONLY END LINE TO KEEP SAME"
1700 DISP "1=SINGLE LOOP"
1710 DISP "2=COINCIDENT LOOP"
1720 DISP "3=CENTRAL INDUCTION LOOP"
1730 DISP "4=OFFSET LOOP" @ INPUT Z$
1740 IF Z$#"" THEN E=VAL(Z$) ELSE E=VAL(E$[2,2])
1750 IF E=1 THEN E$="(1)SINGLE LOOP"
1760 IF E=2 THEN E$="(2)COINCIDENT LOOP"
1770 IF E=3 THEN E$="(3)CENTRAL INDUCTION LOOP"
1780 IF E=4 THEN E$="(4)OFFSET LOOP"
1790 DISP @ DISP F$
1800 DISP "ENTER NEW XMTR LOOP SIDE (m)"
1810 DISP "PRESS ONLY END LINE TO KEEP SAME" @ INPUT Z$
1820 IF Z$#"" THEN F$[13]=Z$
1830 DISP @ DISP G$
1840 IF E#3 THEN DISP "ENTER NEW # OF XMTR LOOP TURNS"
1850 IF E=3 THEN DISP "ENTER NEW RCVR MOMENT(n*A)"
1860 DISP "PRESS ONLY END LINE TO KEEP SAME" @ INPUT Z$
1870 IF Z$#"" THEN G$[15]=Z$
1880 DISP @ DISP "EDIT DATA? Y/N - DEF IS NO" @ INPUT Z$
1890 IF Z$="" THEN RETURN
1900 IF UPC$(Z$[1,1])#"Y" THEN RETURN
1910 FOR I=1 TO N
1920 CLEAR
1930 DISP "SET ";I;" STACKS ";O(I)
1940 DISP "ENTER NEW NUMBER OF STACKS"
1950 DISP "PRESS ONLY END LINE TO KEEP SAME" @ INPUT Z$
1960 IF Z$#"" THEN O(I)=VAL(Z$)
1970 DISP @ DISP "   CHAN   microV/I    SET ";I
1980 DISP "-------------------------------"
1990 CO=0
2000 FOR J=1 TO 32
```

```
2010 IF V(I,J)=0 AND C0>=N(I) THEN 2060
2020 DISP USING 2030 ; J,V(I,J)
2030 IMAGE 4X,2D,2X,6D.3D
2040 IF V(I,J)#0 THEN C0=C0+1
2050 NEXT J
2060 DISP @ DISP "ENTER CHANNEL NUMBER TO EDIT"
2070 DISP "PRESS ONLY END LINE TO CONTINUE" @ INPUT Z$
2080 IF Z$="" THEN 2240
2090 J=VAL(Z$)
2100 IF V(I,J)#0 THEN 2160
2110 DISP @ DISP "ENTER microV/I FOR CHAN ";J
2120 DISP "TO LEAVE EMPTY PRESS END LINE" @ INPUT Z$
2130 IF Z$#"" THEN V(I,J)=VAL(Z$)
2140 IF V(I,J)#0 THEN N(I)=N(I)+1
2150 GOTO 2060
2160 DISP @ DISP "EDIT OLD microV/I = ";V(I,J)
2170 DISP @ DISP "ENTER NEW microV/I"
2180 DISP "PRESS ONLY END LINE TO KEEP SAME"
2190 DISP "TO DELETE ENTER ZERO" @ INPUT Z$
2200 IF Z$="" THEN 2060
2210 V(I,J)=VAL(Z$)
2220 IF V(I,J)=0 THEN N(I)=N(I)-1
2230 GOTO 2060
2240 NEXT I
2250 RETURN
2260 !
2270 ! ADD DATA SETS
2280 !
2290 CLEAR @ CFLAG 1
2300 DISP "ENTER NUMBER OF DATA SETS TO ADD" @ INPUT N0
2310 FOR I=N+1 TO N+N0
2320 CLEAR
2330 DISP "ENTER DATA FOR SET ";I
2340 DISP @ DISP "ENTER NUMBER OF STACKS" @ INPUT O(I)
2350 FOR J=1 TO 32
2360 DISP @ DISP "ENTER CHAN ";J;" microvolts/amp"
2370 DISP "ENTER ZERO TO SKIP CHANNEL"
2380 DISP "PRESS ONLY END LINE TO QUIT" @ INPUT Z$
2390 IF Z$="" THEN 2430
2400 V(I,J)=VAL(Z$)
2410 IF V(I,J)#0 THEN N(I)=N(I)+1
2420 NEXT J
2430 NEXT I
2440 N=N+N0
2450 D$[6]=VAL$(N)
2460 RETURN
2470 !
2480 ! DELETE DATA SETS
2490 !
2500 CLEAR @ CFLAG 1
2510 DISP "ENTER DATA SET NUMBER TO DELETE"
```

```
2520 DISP "PRESS ONLY END LINE TO RETURN" @ INPUT Z$
2530 IF Z$="" THEN RETURN ELSE NO=INT(VAL(Z$))
2540 IF NO>N OR NO<1 THEN BEEP @ DISP @ DISP "NO SUCH SET
NUMBER" @ DISP @ GOTO 2510
2550 FOR I=NO TO N-1
2560 O(I)=O(I+1)
2570 N(I)=N(I+1)
2580 FOR J=1 TO 32
2590 V(I,J)=V(I+1,J)
2600 NEXT J
2610 NEXT I
2620 O(N),N(N)=0
2630 FOR J=1 TO 32
2640 V(N,J)=0
2650 NEXT J
2660 N=N-1
2670 D$[6]=VAL$(N) @ GOTO 2500
2680 RETURN
2690 !
2700 ! DATA STORAGE
2710 !
2720 CLEAR @ CFLAG 1
2730 ON ERROR GOTO 3840
2740 DISP "ENTER OUTPUT FILE NAME" @ INPUT N$
2750 DISP @ DISP "NEW OR OLD NAME? N/O"
2760 DISP "IF OLD, FILE WILL BE REPLACED"
2770 INPUT Z$
2780 IF UPC$(Z$[1,1])="N" THEN 2830
2790 IF UPC$(Z$[1,1])#"O" THEN 2750
2800 DISP @ DISP "PURGE FILE ";N$;"? Y/N" @ INPUT Z$
2810 IF UPC$(Z$[1,1])="Y" THEN PURGE N$
2820 IF UPC$(Z$[1,1])#"Y" THEN 2750
2830 NO=CEIL((169+N*34*8)/256)
2840 CREATE N$,NO,256
2850 ASSIGN# 1 TO N$
2860 CRT OFF
2870 PRINT# 1 ; A$,B$,C$,D$,E$,F$,G$
2880 FOR I=1 TO N
2890 MAT VO=V(I,1:32)
2900 PRINT# 1 ; N(I),O(I),VO()
2910 NEXT I
2920 ASSIGN# 1 TO *
2930 OFF ERROR
2940 CRT ON
2950 RETURN
2960 !
2970 ! PRINT DATA
2980 !
2990 CLEAR @ CFLAG 1
3000 CRT OFF
3010 PRINT
```

```
3020 PRINT A$
3030 PRINT B$
3040 PRINT C$
3050 PRINT D$
3060 PRINT E$
3070 PRINT F$
3080 PRINT G$
3090 PRINT
3100 FOR I=1 TO N
3110 PRINT
3120 PRINT "SET ";I;" STACKS ";O(I)
3130 PRINT
3140 PRINT "   CHAN    microV/I"
3150 PRINT "-------------------------------"
3160 C0=0
3170 FOR J=1 TO 32
3180 IF V(I,J)=0 AND C0>=N(I) THEN 3230
3190 PRINT USING 3200 ; J,V(I,J)
3200 IMAGE 4X,2D,3X,6D.3D
3210 IF V(I,J)#0 THEN C0=C0+1
3220 NEXT J
3230 PRINT
3240 NEXT I
3250 PRINT @ PRINT @ PRINT @ PRINT
3260 CRT ON
3270 RETURN
3280 !
3290 ! HELP ROUTINE
3300 !
3310 CLEAR @ CFLAG 1
3320 DISP "PRINT OR DISP INFORMATION? P/D"
3330 DISP "DEFAULT IS DISP" @ INPUT Z$
3340 IF Z$="" THEN Z$[1]="D" @ GOTO 3360
3350 IF UPC$(Z$[1,1])="P" THEN CRT IS 2 ELSE CRT IS 1
3360 CLEAR @ KEY LABEL
3370 DISP "THE SPECIAL FUNCTION KEYS ARE"
3380 DISP "LABELED AT THE BOTTOM OF THE"
3390 DISP "HP-85 CRT SCREEN. UPPER LABELS"
3400 DISP "REFER TO SHIFTED KEYS AND"
3410 DISP "LOWER LABELS REFER TO UNSHIFTED"
3420 DISP "KEYS. EACH KEY, WHEN PRESSED,"
3430 DISP "WILL PERFORM THE FOLLOWING"
3440 DISP "DEFINED FUNCTIONS:"
3450 DISP
3460 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
3470 CLEAR @ KEY LABEL
3480 DISP "HELP - AN ON LINE HELP LIBRARY"
3490 DISP "EXPLAINING PROGRAM OPTIONS."
3500 DISP @ DISP "DATA - ROUTINE TO ENTER UP TO 10"
3510 DISP "SETS OF TDEM DATA FROM EITHER"
```

```
3520 DISP "THE KEYBOARD OR TAPE STORAGE."
3530 DISP "HEADER INFORMATION AND UP TO 32"
3540 DISP "DATA CHAN/SET MAY BE ENTERED."
3550 DISP
3560 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
3570 CLEAR @ KEY LABEL
3580 DISP "READ - ROUTINE TO RECOVER"
3590 DISP "COMBINED HEADER AND DATA FROM"
3600 DISP "STORAGE FOR EDITING"
3610 DISP @ DISP "EDIT - ROUTINE TO EDIT TDEM"
3620 DISP "HEADER AND DATA. INDIVIDUAL DATA"
3630 DISP "CHANNELS MAY BE SELECTED AND"
3640 DISP "MODIFIED OR DELETED."
3650 DISP
3660 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
3670 CLEAR @ KEY LABEL
3680 DISP "ADD - ROUTINE TO ADD MORE SETS"
3690 DISP "TO A PREVIUSLY CREATED DATA BASE(MAXIMUM OF 10)."
3700 DISP @ DISP "DEL - ROUTINE TO DELETE SETS"
3710 DISP "FROM THE CURRENT DATA BASE."
3720 DISP
3730 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
3740 CLEAR @ KEY LABEL
3750 DISP "STORE - ROUTINE TO STORE TDEM"
3760 DISP "HEADER AND DATA FOR FUTURE      PROCESSING."
3770 DISP @ DISP "PRINT - ROUTINE TO PRINT HEADER"
3780 DISP "AND DATA ON THE HP-85 PRINTER."
3790 DISP
3800 IF UPC$(Z$[1,1])="D" THEN DISP "(PRESS CONTINUE)" @
PAUSE
3810 CRT IS 1
3820 RETURN
3830 !
3840 ! MASS STORAGE ERROR
3850 !
3860 OFF ERROR @ CRT ON @ BEEP
3870 DISP @ DISP "ERROR NUMBER ";ERRN
3880 DISP "ERROR LINE ";ERRL @ ERRM
3890 DISP @ DISP "CORRECT AND CONTINUE WHEN READY" @ PAUSE
3900 GOTO 200 ! RESTART
```

Appendix IV. Program listing for "TDRES".

```
10 ! ******* TDRES ***********
20 ! * TDEM APP. RESISTIVITY  *
30 ! * CALCULATION PROGRAM.   *
40 ! * DATA IS ENTERED FROM   *
50 ! * HP-85 TAPE OR DISK.    *
60 ! * OPTIONS TO PRINT AND   *
70 ! * PLOT RESULTS ON SEVERAL*
80 ! * OUTPUT DEVICES.        *
90 ! *************************
100 ! BY PAUL RAAB AND FRANK FRISCHKNECHT - USGS
110 ! MODIFIED 4/13/83
120 OPTION BASE 1
130 DEG
140 CRT IS 1 @ PRINTER IS 2
150 INTEGER O(10),N(10)
160 SHORT T(32),V(10,32),VO(32),R(10,32)
170 DIM A$[32],B$[8],C$[32],D$[8],E$[25],F$[18],G$[25]
180 DIM NO$[9],N1$[10],I$[96]
190 DIM L1$[4],L2$[7],L3$[9],L4$[11],L5$[23],S$[1]
200 DIM Z$[32],O$[150]
210 !
220 ! SIROTEM TIME VALUES (msec)
230 !
240 DATA .4, .8, 1.2, 1.6, 2, 2.6, 3.4, 4.2, 5, 5.8, 7, 8.6, 10.2, 11.8
250 DATA 13.4, 15.8, 19, 22.2, 25.4, 28.6, 33.4, 39.8, 46.2, 52.6, 59
260 DATA 68.6, 81.4, 94.2, 107, 119.8, 139, 164.6
270 MAT READ T ! READ TIME VALUES
280 !
290 ! ERROR FUNCTION DEFINITION
300 ! FOR CENTRAL LOOP CASE .2<Z<=4
310 ! FROM ABRAMOWITZ & STEGUN, 1965
320 !
330 DEF FNF(Z)
340 U=1/(1+.3275911*Z)
350 U1=U*(.254829592+U*(-.284496736+U*1.421413741))
360 U1=U1+U^4*(-1.453152027+U*1.061405429)
370 U1=3-(3*U1+(3*Z+2*Z^3)*2/SQR(PI))*EXP(-(Z^2))
380 FNF=Z^(-2)*U1
390 FN END
400 !
410 ! OPTION SELECT
420 !
430 CLEAR @ BEEP 200,100
440 DISP "PRINT RESISTIVITY DATA? Y/N"
450 DISP "DEFAULT IS YES" @ INPUT Z$
460 IF Z$="" THEN F0=1 @ GOTO 480
470 IF UPC$(Z$[1])#"N" THEN F0=1 ELSE F0=0 @ GOTO 560
480 DISP @ DISP "ENTER PRINTER ADDRESS"
```

```
490 DISP "1=HP-85 CRT"
500 DISP "2=HP-85 PRINTER"
510 DISP "3=HP-82905A"
520 DISP "DEFAULT=1" @ INPUT Z$
530 IF Z$="" THEN D1=1 @ GOTO 560
540 D1=VAL(Z$)
550 IF D1=3 THEN D1=701 ! SET EXTERNAL PRINTER ADDRESS
560 DISP @ DISP "PLOT DATA? Y/N"
570 DISP "DEFAULT IS YES" @ INPUT Z$
580 IF Z$="" THEN F1=1 @ GOTO 600
590 IF UPC$(Z$[1])#"N" THEN F1=1 ELSE F1=0 @ GOTO 850
600 DISP @ DISP "ENTER PLOTTER ADDRESS"
610 DISP "1=HP-85 CRT"
620 DISP "2=HP-85 PRINTER"
630 DISP "3=HP-7225A"
640 DISP "DEFAULT=1" @ INPUT Z$
650 IF Z$="" THEN D2=1 @ GOTO 680
660 D2=VAL(Z$)
670 IF D2=3 THEN D2=708 ! SET EXTERNAL PLOTTER ADDRESS
680 DISP @ DISP "ENTER PLOTTING OPTION"
690 DISP "1=VECTORS"
700 DISP "2=SYMBOLS"
710 DISP "3=BOTH"
720 DISP "DEFAULT=3" @ INPUT Z$
730 IF Z$="" THEN F2=3 @ GOTO 750
740 F2=VAL(Z$)
750 DISP @ DISP "ENTER SCALING OPTION"
760 DISP "1=AUTO SCALE"
770 DISP "2=USER SELECTED SCALES"
780 DISP "DEFAULT=1" @ INPUT Z$
790 IF Z$="" THEN F3=1 @ GOTO 850
800 F3=VAL(Z$)
810 IF F3=1 THEN 850
820 DISP @ DISP "ENTER MIN,MAX LOG TIME" @ INPUT T1,T2
830 DISP @ DISP "ENTER MIN,MAX LOG TRANSIENT" @ INPUT V1,V2
840 DISP @ DISP "ENTER MIN,MAX LOG APP. RES." @ INPUT R1,R2
850 DISP @ DISP "ENABLE EARLY TIME SWITCH? Y/N"
860 DISP "DEFAULT IS NO" @ INPUT Z$
870 IF Z$="" THEN F4=0 @ GOTO 900
880 IF UPC$(Z$[1])="Y" THEN F4=1 ELSE F4=0
890 !
900 ! GET INPUT FILE NAMES
910 !
920 CLEAR
930 DISP "ENTER GENERAL INPUT FILE NAME" @ INPUT NO$
940 DISP @ DISP "ENTER SELECTED FILE NUMBERS"
950 LINPUT "(e.g. 1,2,3,4 <CR/LF>)          ?",I$
960 P1=1 @ P2=0
970 IF LEN(I$)=0 THEN N1$=TRIM$(NO$) @ GOTO 1010
980 P2=POS(I$[P1],",")+P1-2
990 IF P2>=P1 THEN NO=VAL(I$[P1,P2]) ELSE NO=VAL(I$[P1])
```

```
1000 N1$=TRIM$(N0$)&VAL$(N0)
1010 MAT V=ZER@ MAT R=ZER
1020 ASSIGN# 1 TO N1$
1030 CRT OFF
1040 READ# 1 ; A$, B$, C$, D$, E$, F$, G$
1050 N=VAL(D$[6])
1060 FOR I=1 TO N
1070 READ# 1 ; N(I),O(I),V0()
1080 MAT V(I,1:32)=V0
1090 NEXT I
1100 ASSIGN# 1 TO *
1110 CRT ON @ CLEAR
1120 DISP "INPUT FILE = ";N1$
1130 DISP A$ @ DISP B$ @ DISP C$
1140 DISP D$ @ DISP E$ @ DISP F$ @ DISP G$
1150 L=VAL(F$[13])
1160 M=VAL(G$[15])
1170 GOSUB 1240 ! APP. RESIS.
1180 GOSUB 2410 ! PRINT DATA
1190 GOSUB 3560 ! PLOT DATA
1200 IF P2>=P1 THEN P1=P2+2 @ GOTO 980
1210 CLEAR @ DISP "NORMAL END"
1220 END
1230 !
1240 ! APPARENT RESISTIVITY
1250 ! CALCULATION
1260 !
1270 DISP @ DISP "CALCULATING APPARENT RESISTIVITY"
1280 CFLAG 1
1290 FOR I=1 TO N
1300 IF E$[2,2]#"3" THEN GOSUB 1350 ELSE GOSUB 1610
1310 NEXT I
1320 CLEAR
1330 RETURN
1340 !
1350 ! SINGLE, OFFSET, OR
1360 ! COINCIDENT LOOP
1370 !
1380 ! L=XMTR SIDE(m), M=XMTR TURNS, T(J)=TIME(msec)
1390 ! V(I,J)=V/I(microvolts/amp), R(I,J)=APP RESISTIVITY
1400 !
1410 FOR J=1 TO 32
1420 IF V(I,J)=0 THEN R(I,J)=0 @ GOTO 1580
1430 Y=ABS(V(I,J))*T(J)/(800*PI*L*M*M)
1440 Z=Y^(2/3)
1450 X1=1.70998+Z*(2.38095+Z*(6.49229+Z*20.8835))
1460 X1=X1+Z^4*(71.8975+Z*(255.846+Z*(925.902+Z*3378.09)))
1470 X1=Z/*(X1+Z^8*(12360.9+Z*110000))^2
1480 IF X1<=1.4 THEN X=X1
1490 IF 1.4<X1 AND X1<=2.8 THEN X=X1+.001635*X1^4.892
1500 IF 2.8<X1 AND X1<=5.69 THEN X=X1+.004018*X1^4.01364
```

```
1510 IF X1>5.69 THEN 1520 ELSE 1570
1520 IF FLAG(1) THEN 1540
1530 PRINT @ PRINT "INPUT FILE = ";N1$ @ PRINT @ SFLAG 1
1540 PRINT "NO SOLUTION FOR"
1550 PRINT "SET ";I;" TIME ";T(J);" msec"
1560 PRINT @ GOTO 1580
1570 R(I,J)=.0001*L*L/(T(J)*X)
1580 NEXT J
1590 RETURN
1600 !
1610 ! CENTRAL INDUCTION LOOP
1620 !
1630 ! L=XMTR SIDE(m), M=RCVR MOMENT, T(J)=TIME(msec)
1640 ! V(I,J)=V/I(microvolts/amp), R(I,J)=APP RESISTIVITY
1650 !
1660 F5=0 @ Y3=-1.E99
1670 FOR J=32 TO 1 STEP -1
1680 IF V(I,J)=0 THEN 2380
1690 V(I,J)=ABS(V(I,J))
1700 Y0=.01*T(J)*V(I,J)*L/(2*M*PI)
1710 Y=Y0*2/SQR(PI)
1720 IF Y>.701585 THEN 2320
1730 IF F4=0 THEN 1770
1740 IF Y0<Y3 OR F5=1 THEN 2110 ! EARLY TIME TEST
1750 Y3=Y0
1760 !
1770 ! LATE TIME SIDE
1780 !
1790 U=Y0^(1/3)
1800 Z=U*(1.077217345+U*(U*(.321+U*(-.14+U*.42))))
1810 Z=Z1+U^6*(-.22+U*(.6+U*(-.86+U*(.98-U*.8))))
1820 Z=Z1+U^11*(.68+U*(-.32+U*1.52))
1830 IF Z<=.5 THEN 2370
1840 !
1850 ! ITERATIVE SERIES FOR .5<Z
1860 !
1870 IF Z>1.61362 THEN Z=1.61362
1880 X1=Z1
1890 Y1=FNF(Z)
1900 IF ABS(Y1-Y)<.00005*Y THEN 2370
1910 FOR K=1 TO 6
1920 X2=X1+X1*(K*(1-Y1/Y))
1930 IF X2>1.61362 THEN X2=1.61362
1940 Y2=FNF(X2)
1950 IF ABS(Y2-Y)<.00005*Y THEN Z=X2 @ GOTO 2370
1960 IF Y1<Y AND Y<Y2 THEN 2010
1970 IF Y2<Y AND Y<Y1 THEN 2000
1980 NEXT K
1990 GOTO 2320
2000 X1=X2 @ X2=Z1
2010 FOR K=1 TO 10
```

```
2020 Z=X1+(X2-X1)/2
2030 IF Z>1.61362 THEN Z=1.61362
2040 Y1=FNF(Z)
2050 IF ABS(Y1-Y)<.0005*Y THEN 2370
2060 IF Y1<Y THEN X1=Z
2070 IF Y<Y1 THEN X2=Z
2080 NEXT K
2090 GOTO 2320
2100 !
2110 ! EARLY TIME SIDE
2120 !
2130 F5=1
2140 Z=SQR(3/Y)
2150 IF Z>3 THEN 2370
2160 FOR K=1 TO 3
2170 X1=3/1.2296^K
2180 Y1=FNF(X1)
2190 IF ABS(Y1-Y)<.001*Y THEN 2370
2200 IF Y<Y1 THEN 2230
2210 NEXT K
2220 X1=.88889 @ X2=1.6138 @ GOTO 2240
2230 X1=3/1.2296^K @ X2=3/1.2296^(K-1)
2240 FOR K=1 TO 10
2250 Z=X1+(X2-X1)/2
2260 IF Z<1.6138 THEN Z=1.6138
2270 Y1=FNF(Z)
2280 IF ABS(Y1-Y)<.001*Y THEN 2370
2290 IF Y<Y1 THEN X1=Z
2300 IF Y1<Y THEN X2=Z
2310 NEXT K
2320 IF FLAG(1) THEN 2340
2330 PRINT @ PRINT "INPUT FILE = ";N1$ @ PRINT @ SFLAG 1
2340 PRINT "NO SOLUTION FOR"
2350 PRINT "SET ";I;" TIME ";T(J);" msec"
2360 PRINT @ GOTO 2380
2370 R(I,J)=.0001*L*L/(T(J)*Z^2)
2380 NEXT J
2390 RETURN
2400 !
2410 ! DATA PRINT ROUTINE
2420 !
2430 CLEAR
2440 IF F0=0 THEN RETURN
2450 IF D1>2 THEN GOTO 2740
2460 !
2470 ! HP-85 PRINT
2480 !
2490 PRINTER IS D1
2500 IF D1=2 THEN CRT OFF
2510 PRINT @ PRINT "INPUT FILE = ";N1$
2520 PRINT A$ @ PRINT B$ @ PRINT C$
```

```
2530 PRINT D$ @ PRINT E$ @ PRINT F$ @ PRINT G$ @ PRINT
2540 FOR I=1 TO N
2550 PRINT "SET ";I;" CHAN = ";N(I)
2560 PRINT "STACKS ";O(I)
2570 PRINT
2580 PRINT "CHAN  TIME  microV/I     APP RES"
2590 PRINT "-------------------------------"
2600 C0=0
2610 FOR J=1 TO 32
2620 IF V(I,J)=0 AND C0>=N(I) THEN 2670
2630 PRINT USING 2640 ; J,T(J),V(I,J),R(I,J)
2640 IMAGE 1X,2D,X,3D.D,2X,D.3DE,2X,D.2DE
2650 IF V(I,J)>0 THEN C0=C0+1
2660 NEXT J
2670 PRINT
2680 NEXT I
2690 PRINT @ PRINT
2700 CRT ON
2710 PRINTER IS 2
2720 RETURN
2730 !
2740 ! IMPACT PRINTER ROUTINE
2750 !
2760 PRINTER IS D1,80
2770 CRT OFF
2780 OUTPUT D1 USING "#,K" ; CHR$(27)&"C"&CHR$(80)
2790 OUTPUT D1 USING "#,K" ; CHR$(27)&"C"&CHR$(0)&CHR$(11)
2800 OUTPUT D1 USING "#,K" ; CHR$(27)&"0"
2810 OUTPUT D1 USING "#,K" ; CHR$(27)&"N"&CHR$(8)
2820 PRINT "INPUT FILE = ";TRIM$(N1$)
2830 PRINT TRIM$(A$)&"  "&TRIM$(B$)&"  "&TRIM$(C$)
2840 PRINT TRIM$(E$)&"  "&TRIM$(F$)&"  "&TRIM$(G$) @ PRINT
2850 FOR I=1 TO N STEP 3
2860 L1$="SET " @ L2$="CHAN = " @ L3$="STACKS = "
2870 IF N-I=0 THEN 2990
2880 IF N-I=1 THEN 2940
2890 PRINT USING 2900 ;
L1$,I,L2$,N(I),L1$,I+1,L2$,N(I+1),L1$,I+2,L2$,N(I+2)
2900 IMAGE 13X,2(4A,2D,2X,7A,2D,6X),4A,2D,2X,7A,2D
2910 PRINT USING 2920 ; L3$,O(I),L3$,O(I+1),L3$,O(I+2)
2920 IMAGE 15X,2(9A,4D,10X),9A,4D
2930 GOTO 3030
2940 PRINT USING 2950 ; L1$,I,L2$,N(I),L1$,I+1,L2$,N(I+1)
2950 IMAGE 13X,2(4A,2D,2X,7A,2D,6X)
2960 PRINT USING 2970 ; L3$,O(I),L3$,O(I+1)
2970 IMAGE 15X,2(9A,4D,10X)
2980 GOTO 3030
2990 PRINT USING 3000 ; L1$,I,L2$,N(I)
3000 IMAGE 13X,4A,2D,2X,7A,2D
3010 PRINT USING 3020 ; L3$,O(I)
3020 IMAGE 15X,9A,4D
```

```
3030 PRINT
3040 L4$="  CH   T(ms)"
3050 L5$="     microV/I     APP. RES."
3060 IF N-I>=2 THEN PRINT L4$&L5$&L5$&L5$
3070 IF N-I=1 THEN PRINT L4$&L5$&L5$
3080 IF N-I=0 THEN PRINT L4$&L5$
3090 L4$="  --   -----"
3100 L5$="  ----------   ---------"
3110 IF N-I>=2 THEN PRINT L4$&L5$&L5$&L5$
3120 IF N-I=1 THEN PRINT L4$&L5$&L5$
3130 IF N-I=0 THEN PRINT L4$&L5$
3140 M0=N-I
3150 IF M0>=2 THEN M0=2
3160 N1=-INF
3170 FOR J0=0 TO M0
3180 FOR J=1 TO 32
3190 IF V(I+J0,J)=0 THEN 3210
3200 N1=MAX(N1,J)
3210 NEXT J
3220 NEXT J0
3230 FOR J=1 TO N1
3240 O$=""
3250 IF J<10 THEN O$[1,2]="  " @ O$[3,3]=VAL$(J)
3260 IF J>=10 THEN O$[1,1]=" " @ O$[2,3]=VAL$(J)
3270 IF T(J)<1 THEN O$[4,7]="     " @ O$[8,9]=VAL$(T(J))
3280 IF T(J)>=1 AND T(J)<10 THEN O$[4,6]="    " @
O$[7,9]=VAL$(T(J))
3290 IF T(J)>=10 AND T(J)<100 THEN O$[4,5]="  " @
O$[6,9]=VAL$(T(J))
3300 IF T(J)>=100 THEN O$[4,4]=" " @ O$[5,9]=VAL$(T(J))
3310 O$[10,10]=" "
3320 FOR K=0 TO M0
3330 P3=K*23+11
3340 P4=P3+14
3350 IF V(I+K,J)=0 THEN E0=0 ELSE E0=INT(LGT(V(I+K,J)))
3360 E1=V(I+K,J)/10^E0
3370 IF E0<0 THEN E0=ABS(E0) @ S$="-" ELSE S$="+"
3380 O$[P3,P3+6]="  "&VAL$(E1)
3390 O$[P3+7,P3+11]="E"&S$&"00"&VAL$(E0)
3400 O$[P3+12,P3+13]="  "
3410 IF R(I+K,J)=0 THEN E0=0 ELSE E0=INT(LGT(R(I+K,J)))
3420 E1=R(I+K,J)/10^E0
3430 IF E0<0 THEN E0=ABS(E0) @ S$="-" ELSE S$="+"
3440 O$[P4,P4+3]=VAL$(E1)
3450 O$[P4+4,P4+8]="E"&S$&"00"&VAL$(E0)
3460 NEXT K
3470 PRINT O$
3480 NEXT J
3490 PRINT
3500 NEXT I
3510 PRINT CHR$(12) ! FORM-FEED
```

```
3520 CRT ON
3530 PRINTER IS 2
3540 RETURN
3550 !
3560 ! DATA PLOT ROUTINE
3570 !
3580 CLEAR
3590 IF F1=0 THEN RETURN
3600 !
3610 ! MAX/MIN ROUTINE
3620 !
3630 IF F3=2 THEN 3810
3640 V1,R1,T1=INF @ V2,R2,T2=-INF
3650 FOR I=1 TO N
3660 FOR J=1 TO 32
3670 IF V(I,J)=0 THEN 3750
3680 V1=MIN(V1,V(I,J))
3690 V2=MAX(V2,V(I,J))
3700 IF R(I,J)=0 THEN 3750
3710 R1=MIN(R1,R(I,J))
3720 R2=MAX(R2,R(I,J))
3730 T1=MIN(T1,T(J))
3740 T2=MAX(T2,T(J))
3750 NEXT J
3760 NEXT I
3770 V1=FLOOR(LGT(V1)) @ V2=CEIL(LGT(V2))
3780 R1=FLOOR(LGT(R1)) @ R2=CEIL(LGT(R2))
3790 T1=FLOOR(LGT(T1)) @ T2=CEIL(LGT(T2))
3800 !
3810 ! HP-85 PLOTTER SUBROUTINE
3820 !
3830 IF D2>2 THEN 4640
3840 PLOTTER IS 1
3850 X1=T1 @ X2=T2
3860 FOR K=1 TO 2
3870 GCLEAR
3880 IF D2=1 THEN 3930
3890 PRINT @ PRINT
3900 IF K=1 THEN PRINT "            LOG TRANSIENT"
3910 IF K=2 THEN PRINT "         LOG APP RESISTIVITY"
3920 IF D2=2 THEN PRINT
3930 LOCATE 10,122,15,95
3940 IF K=1 THEN Y1=V1 @ Y2=V2 ELSE Y1=R1 @ Y2=R2
3950 SCALE X1,X2,Y1,Y2
3960 FRAME
3970 !
3980 ! DRAW LOG CYCLES
3990 !
4000 FOR I=X1 TO X2-1
4010 FOR J=1 TO 10
4020 MOVE I+LGT(J),Y1 @ SETGU @ IDRAW 0,1 @ SETUU @ PENUP
```

```
4030 NEXT J
4040 NEXT I
4050 FOR I=X1 TO X2-1
4060 FOR J=1 TO 10
4070 MOVE I+LGT(J),Y2 @ SETGU @ IDRAW 0,-1 @ SETUU @ PENUP
4080 NEXT J
4090 NEXT I
4100 FOR I=Y1 TO Y2-1
4110 FOR J=1 TO 10
4120 MOVE X1,I+LGT(J) @ SETGU @ IDRAW 1,0 @ SETUU @ PENUP
4130 NEXT J
4140 NEXT I
4150 FOR I=Y1 TO Y2-1
4160 FOR J=1 TO 10
4170 MOVE X2,I+LGT(J) @ SETGU @ IDRAW -1,0 @ SETUU @ PENUP
4180 NEXT J
4190 NEXT I
4200 !
4210 ! LABEL AXES
4220 !
4230 LDIR 0 @ LORG 4
4240 SETGU @ MOVE 60,0 @ LABEL "LOG TIME" @ SETUU
4250 FOR J=X1 TO X2
4260 MOVE J,Y1 @ SETGU @ IMOVE 0,-6 @ LABEL J @ SETUU
4270 NEXT J
4280 LORG 1
4290 FOR J=Y1 TO Y2
4300 MOVE X1,J @ SETGU @ IMOVE -10,0 @ LABEL J @ SETUU
4310 NEXT J
4320 !
4330 ! PLOT DATA
4340 !
4350 FOR I=1 TO N
4360 IF F2=2 THEN 4430
4370 FOR J=1 TO 32
4380 IF K=1 AND V(I,J)=0 THEN 4420
4390 IF K=2 AND R(I,J)=0 THEN 4420
4400 IF K=1 THEN PLOT LGT(T(J)),LGT(V(I,J))
4410 IF K=2 THEN PLOT LGT(T(J)),LGT(R(I,J))
4420 NEXT J
4430 PENUP
4440 LORG 5
4450 IF F2=1 THEN 4570
4460 FOR J=1 TO 32
4470 IF LGT(T(J))<T1 OR LGT(T(J))>T2 THEN 4560
4480 IF K=2 THEN 4530
4490 IF V(I,J)=0 THEN 4560
4500 IF LGT(V(I,J))<V1 OR LGT(V(I,J))>V2 THEN 4560
4510 MOVE LGT(T(J)),LGT(V(I,J)) @ LABEL "X"
4520 GOTO 4560
4530 IF R(I,J)=0 THEN 4560
```

```
4540 IF LGT(R(I,J))<R1 OR LGT(R(I,J))>R2 THEN 4560
4550 MOVE LGT(T(J)),LGT(R(I,J)) @ LABEL "O"
4560 NEXT J
4570 PENUP
4580 NEXT I
4590 IF D2=2 THEN GRAPH @ COPY
4600 NEXT K
4610 IF D2=2 THEN PRINT @ PRINT @ PRINT @ PRINT
4620 RETURN
4630 !
4640 ! EXTERNAL PLOT SETUP
4650 !
4660 DISP "LOAD PAPER AND CONTINUE" @ BEEP 200,100 @ PAUSE
4670 PLOTTER IS D2
4680 !
4690 ! LABEL TITLES
4700 !
4710 LOCATE 0,140,0,100
4720 CSIZE 4 @ LORG 6
4730 MOVE 70,100
4740 LABEL TRIM$(A$)&"  "&TRIM$(B$)&"  "&TRIM$(C$)
4750 LABEL TRIM$(E$)&"  "&TRIM$(F$)&"  "&TRIM$(G$)
4760 MOVE 70,0 @ LORG 4
4770 LABEL "LOG TIME(msecs)"
4780 LOCATE 12,125,10,90
4790 FRAME
4800 !
4810 ! SCALE
4820 !
4830 FOR K=1 TO 2
4840 LORG 6 @ SETGU @ CSIZE 4
4850 IF K=1 THEN LDIR 90 @ MOVE 0,50
4860 IF K=1 THEN LABEL "LOG TRANSIENT(microvolts/amp)"
4870 IF K=2 THEN LDIR 270 @ MOVE 138,50
4880 IF K=2 THEN LABEL "LOG APP. RESISTIVITY(ohm-meters)"
4890 X1=T1 @ X2=T2
4900 IF K=1 THEN Y1=V1 @ Y2=V2 ELSE Y1=R1 @ Y2=R2
4910 SCALE X1,X2,Y1,Y2
4920 !
4930 ! LABEL X-AXIS
4940 !
4950 IF K=2 THEN 5120
4960 PENUP @ LDIR 0
4970 FOR I=X1 TO X2
4980 CSIZE 4
4990 MOVE I,Y1 @ SETGU @ IMOVE 0,-5 @ LORG 4 @ LABEL "10"
5000 IMOVE 2,6 @ LORG 1 @ CSIZE 3 @ LABEL I @ SETUU
5010 IF I=X2 THEN 5060
5020 FOR J=1 TO 10
5030 MOVE I+LGT(J),Y1 @ SETGU @ IDRAW 0,1.5 @ SETUU @ PENUP
5040 NEXT J
```

```
5050 NEXT I
5060 FOR I=X1 TO X2-1
5070 FOR J=1 TO 10
5080 MOVE I+LGT(J),Y2 @ SETGU @ IDRAW 0,-1.5 @ SETUU @ PENUP
5090 NEXT J
5100 NEXT I
5110 !
5120 ! LABEL Y-AXIS
5130 !
5140 LDIR 0 @ LORG 1
5150 FOR I=Y1 TO Y2
5160 IF K=1 THEN MOVE X1,I @ SETGU @ IMOVE -7,0
5170 IF K=2 THEN MOVE X2,I @ SETGU @ IMOVE 2,0
5180 CSIZE 4 @ LABEL "10"
5190 IMOVE 3,6 @ CSIZE 3 @ LABEL I @ SETUU
5200 IF I=Y2 THEN 5280
5210 FOR J=1 TO 10
5220 IF K=1 THEN MOVE X1,I+LGT(J) @ SETGU @ IDRAW 1.5,0
5230 IF K=2 THEN MOVE X2,I+LGT(J) @ SETGU @ IDRAW -1.5,0
5240 SETUU @ PENUP
5250 NEXT J
5260 NEXT I
5270 !
5280 ! DRAW VECTORS
5290 !
5300 FOR I=1 TO N
5310 IF F2=2 THEN 5390
5320 IF K=1 THEN LINETYPE 1 ELSE LINETYPE 4,2
5330 FOR J=1 TO 32
5340 IF K=1 AND V(I,J)=0 THEN 5380
5350 IF K=2 AND R(I,J)=0 THEN 5380
5360 IF K=1 THEN PLOT LGT(T(J)),LGT(V(I,J))
5370 IF K=2 THEN PLOT LGT(T(J)),LGT(R(I,J))
5380 NEXT J
5390 PENUP @ LINETYPE 1
5400 !
5410 ! DRAW SYMBOLS
5420 !
5430 IF F2=1 THEN 5560
5440 LORG 5 @ LDIR 0 @ CSIZE 3
5450 FOR J=1 TO 32
5460 IF LGT(T(J))<X1 OR LGT(T(J))>X2 THEN 5550
5470 IF K=2 THEN 5520
5480 IF V(I,J)=0 THEN 5550
5490 IF LGT(V(I,J))<V1 OR LGT(V(I,J))>V2 THEN 5550
5500 MOVE LGT(T(J)),LGT(V(I,J)) @ LABEL "X"
5510 GOTO 5550
5520 IF R(I,J)=0 THEN 5550
5530 IF LGT(R(I,J))<R1 OR LGT(R(I,J))>R2 THEN 5550
5540 MOVE LGT(T(J)),LGT(R(I,J)) @ LABEL "O"
5550 NEXT J
```

```
5560 PENUP
5570 NEXT I
5580 PENUP
5590 NEXT K
5600 RETURN
```